1 O Movimento dos Corpos

1.3 Gotas de chuva e pára-quedistas (v'(t) = - g - rv(t))

Em ambos os casos trata-se de objetos que caem de grandes alturas e que são freados pela resistência aerodinâmica. A forma precisa da força de arrasto depende de muitos fatores.

Para facilitar a nossa vida, fazemos a suposição de que a aceleração a seja uma função linear de v, isto é:

$$\mathbf{a = dv/dt = -g - r \cdot v(t)} \qquad (1.3\text{-}1)$$

O termo **-**rv(t) -multiplicado pela massa m do corpo- representa a força de arrasto. O termo **-**m**g** seria a força gravitacional que atua junto com a força de arrasto sobre o corpo. $\mathbf{F}_g$ = -m**g** está dirigida para baixo. A força de arrasto está dirigida para cima e v é negativa, pois vamos orientar o eixo y também para cima.

Equação (1.3-1) é só válida para um "sky diver" com o pára-quedas fechado. O corpo começa cair com velocidade zero, e só após do começo do movimento se desenvolve uma força de atrito. Esta força de arrasto (atrito) vai aumentar até equilibrar a força gravitacional. O corpo cai agora com uma **velocidade terminal** constante, pois a aceleração a passa a ser zero. Para determinar a velocidade e a altura do objeto, utilizamos o programa *MuPAD*. Devemos realizar as seguintes etapas:

- ```
  //g:=9.81:
  //v0:=0:
  vel:=ode({v'(t)=-g-r*v(t), v(0)=v0},v(t)):
  velocidade:=solve(vel)
  ```

$$\left\{ -\frac{g - r \cdot e^{-r \cdot \left(t - \frac{\ln\left(\frac{g + r \cdot v0}{r}\right)}{r}\right)}}{r} \right\}$$
  ```

```
simplify(op(%))
```

$$\frac{g \cdot e^{-r \cdot t} - g + r \cdot v0 \cdot e^{-r \cdot t}}{r}$$

- `expand(%)`

$$\frac{v0}{e^{r \cdot t}} - \frac{g}{r} + \frac{g}{r \cdot e^{r \cdot t}}$$

Este resultado vamos simplificar manualmente:

$$v = dy/dt = -g/r + (v_0 + g/r)e^{-rt} \qquad (1.3\text{-}2)$$

Para obter a altura, resolvemos a Equação Diferencial (1.3-1) com v(t) = y'(t) e dv/dt = y"(t):

- `alt:=ode({y''(t)=-g-r*y'(t),y'(0)=v0,y(0)=y0},y(t)):`

  `altura := solve(alt)`

$$\left\{\frac{g + r \cdot (v0 + r \cdot y0) - g \cdot r \cdot t}{r^2} - \frac{e^{-r \cdot t} \cdot (g + r \cdot v0)}{r^2}\right\}$$

É útil escrever esta solução formidável na seguinte forma quase inócua:

$$y = y_0 - gt/r + (rv_0 + g)(1-e^{-rt})/r^2 \qquad (1.3\text{-}3)$$

Da equação (1.3-2) concluímos que a velocidade do corpo vai diminuir com o tempo. Depois de um intervalo de tempo suficientemente longo, v será, aproximadamente, igual à velocidade terminal $\mathbf{v_{term} = -g/r}$.
Para uma gota de chuva (raio = 1,5 mm), temos $v_{term} \approx 7$ m/s. Para atingir esta velocidade, a gota precisa cair apenas cerca de 6 metros.
Um "Sky diver" alcança sua velocidade terminal (≈ 50 m/s) após de uma queda de aproximadamente 450 m.
Com estes dados obtemos um fator de atrito de r ≈ 10 ms^{-1}/50 ms^{-1} = 0,2 para o "Sky diver". No caso de um pingo de chuva temos r ≈ 10/7 = 1,4, então bastante grande.

Qual seria para o "sky diver" o tempo para alcançar o valor da velocidade terminal, se não existisse a força de resistência do ar (= queda livre)?

Resposta:

Sabemos que no caso hipotético da queda livre v = -gt. O seja, t = 50/10 s = 5 s.

Este tempo puramente teórico podemos chamar de *constante de tempo* do "Sky diver". Este tempo virtual podemos denominar com o mesmo símbolo τ que foi usado na seção 1.2. Temos, então, a relação τ = **1/r.**
Agora fazemos uma representação de v para a queda de um pára-quedista caindo de uma altura de 1500 m.
Traçemos os gráficos de v (com ar e sem ar) e da reta da velocidade terminal para os primeiros 8 segundos. (Vamos traçar os valores absolutos das velocidades.)

```
• v:=t->-g/r+(v0+g/r)*exp(-r*t):// vel. com resistência do ar
  vt:=t->-g/r://velocidade terminal
  vl:=t->-g*t://velocidade para a queda livre
  vel:=plot::Function2d(-v(t), t=0..8,Color=RGB::Green):
  gt:=plot::Function2d(-vt(t), t=0..8,Color=RGB::Blue):
  gl:=plot::Function2d(-vl(t), t=0..8,Color=RGB::Red):
  plot(vel,gt,gl,
  AxesTitles=["t/s","vel(m/s)"],Header="Pára-quedista")
```



Fig. 1.3- 1

1.3.1 Solução com lápis e papel

É dada a **E**quação **D**iferencial (1.3-1):

$$a = dv/dt = - g - rv(t) \qquad (1.3\text{-}1)$$

Para resolvê-la, procuramos a função v(t) tal, que quando v(t) e sua primeira derivada forem substituídas na equação (1.3-1), a equação seja satisfeita.
A técnica a aplicar, é o método da separação de variáveis: dv/(rv+g) = -dt.
Cada membro da igualdade possui, agora, apenas uma variável e assim podemos realizar a integração facilmente.
Integrando ambos os membros da igualdade, o seja, escrever $\int dv/(rv+g) = -\int dt + C_1$, obteremos $\ln(rv+g)/r = -t + C_1$.
As condições iniciais bastam para determinar o valor da constante de integração C_1.
Para t = 0, temos $v = v_0$ e para a constante de integração obteremos o valor $C_1=\ln(rv_0+g)/r$.
Façamos alguns transformações adicionais:

$$rv + g = e^{-rt+\ln(rv_0+g)} = (rv_0 + g)e^{-rt}$$

Agora, para a velocidade temos

$$v = \frac{dy}{dt} = -\frac{g}{r} + (v_0 + \frac{g}{r})e^{-rt} \qquad (1.3\text{-}4)$$

Se orientarmos o eixo y para baixo, então devemos escrever -g em vez de g. Caso v =0 para t = 0, obtemos simplesmente

$$v(t) = \frac{g}{r}(1 - e^{-rt}) \qquad (1.3\text{-}5)$$

Agora, integrando a equação (1.3-4), dará para a posição y do corpo as seguintes expressões:

$$y = \int -\frac{g}{r}dt + (v_0 + \frac{g}{r})\int e^{-rt}dt + C_2$$

$$y = -\frac{g}{r}t - \frac{rv_0 + g}{r^2}e^{-rt} + C_2$$

A posição para t = 0 é $y(0) = y_0$ o que dá para C_2 a expressão $C_2 = y_0 + (rv_0 + g)/r^2$.

Assim, temos, finalmente,

$$y(t) = y_0 - \frac{g}{r}t + \frac{rv_0 + g}{r^2}(1 - e^{-rt}) \qquad (1.3\text{-}6)$$

Muito bem. Caso você encontrou algum problema, não entre em pânico, fale, como sempre, com *MuPAD*.

Antes de continuar com esta matéria , quero adicionar algumas observações:

A equação (1.3-1), ou na forma mais geral de $y'(t) = dy/dt = a - by(t)$, é um modelo matemático para muitos processos na natureza.

A equação para a **desintegração radioativa** tem a forma $dN/dt = a - bN$ onde N é o número de núcleos radioativos e a é a taxa de produção do nuclídeo radioativo. O termo bN representa a taxa de decaimento e b é a constante de desintegração (o inverso da meia-vida de um nuclídeo radioativo multiplicado pelo logaritmo neperiano de dois). Para a = 0 e $N(0) = N_0$ resulta $N(t) = N_0e^{-bt}$

Outro exemplo é um **circuito de malha simples**, contendo um resistor R e um indutor L, que é conectado no tempo t = 0 com uma bateria com uma diferença de potencial U. A regra das malhas nos dá para a corrente I a seguinte equação

$$\frac{dI}{dt} = \frac{U}{L} - \frac{R}{L}I(t) \qquad (1.3\text{-}7)$$

Esta equação é uma equação diferencial, envolvendo a variável I e sua primeira derivada dI/dt, que possui exatamente a forma da equação $y'(t) = dy/dt = a - by(t)$. A solução deve ter exatamente a mesma forma da equação (1.3-5), o seja:

$$I(t) = \frac{U}{R}(1 - e^{-bt}) \qquad (1.3\text{-}8)$$

Se substituirmos t = 0 na equação (1.3-8), a exponencial se torna $e^{-0} = 1$, o seja, (1.3-8) nos diz que a corrente era inicialmente zero,como esperávamos. Se fizermos t se aproximar de ∞ , então a exponential tende a zero e a corrente tende ao seu valor limite de $I_{inf} = U/R$.
Após de $\tau = 1/b = L/R$ segundos (= constante de tempo indutiva), a corrente atinge cerca de 63% do seu valor final.

Vejamos o gráfico da corrente crescente com o tempo para os valores de R = 2000 Ω , L = 4 H e U = 10 V.

```
R:=2000: L:=4:U:=10:
i:=t->U*(1-exp(-R*t/L))/R:
curva:=plot::Function2d(i(t),t=0..0.01):
plot(curva,
AxesTitles = ["t/s","I/A"],
Header="I(t) no circuito-R-L simples",
GridVisible=TRUE)
```



Fig. 1,3-2

A corrente atinge depois de, aproximadamente, 10 ms o seu valor limite de 5 mA. Depois de 4/2000 s = 2 ms, a corrente alcança o valor 0.63 · 10/2000 s = 3.2 mA.